Algorithm AS 149

Amalgamation of Means in the Case of Simple Ordering

By G. W. Cran

The Queen's University of Belfast, N. Ireland

Keywords: INFERENCE UNDER SIMPLE ORDERING; ISOTONIC REGRESSION

Language

ISO Fortran

Description and Purpose

Given a set of values of a function with a corresponding set of positive weights, the algorithm computes the values of the associated isotonic regression function with respect to simple ordering. The algorithm is based on the "Up-and-Down Blocks" algorithm developed by J. B. Kruskal and described by means of a flow chart by Barlow *et al.* (1972, p. 73).

The algorithm has applications in isotonic regression when the ordering is simple; in the maximum likelihood estimation of the means $\mu_1, ..., \mu_k$ of the normal distributions $N(\mu_i, \sigma^2)$ $(i = 1, ..., k)$ subject to $\mu_1 \leqslant \mu_2 \leqslant ... \leqslant \mu_k$; in the calculation of Bartholomew's $\bar{E}_k^2$ test statistic for testing homogeneity of means against the simply ordered alternative; and in multidimensional scaling.

Structure

SUBROUTINE AMALGM (*K*, *XO*, *WO*, *X*, *W*, *XA*, *IFAULT*)

Formal parameters

K	Integer	input:	the number of means
XO	Real array (*K*)	input:	the original means
WO	Real array (*K*)	input:	the original weights, all positive
X	Real array (*K*)	workspace	
W	Real array (*K*)	workspace	
XA	Real array (*K*)	output:	the amalgamated means
IFAULT	Integer	output:	a fault indicator, equal to 1 if $K<2$; 2 if any element of *WO* is not positive; 0 otherwise

Data constant TOL:

In the main section of the algorithm, the working weights $\{W(I)\}$ are obtained by addition of a subset of the original weights:

$$W(I) = WO(I1+1)+...+WO(I1+L).$$

In the final section the number L is obtained by comparing $W(I)$ with the partial sum

$$S = WO(I1+1)+...+WO(I1+J) \quad \text{for } J = 1, ...$$

until $W(I)$ equals S whence $L = J$. Because of the possible effects of rounding-off error, the direct test of equality of $W(I)$ and S is replaced by the test of $ABS(W(I)-S)$. $LT.$ TOL.

Acknowledgement

I am grateful to the editor for his helpful comments.

Reference

Barlow, R. E., Bartholomew, D. J., Bremner, J. M. and Brunk, H. D. (1972). *Statistical Inference under Order Restrictions.* New York: Wiley.

```
      SUBROUTINE AMALGM(K, XO, WO, X, W, XA, IFAULT)
C
C        ALGORITHM AS 140 APPL. STATIST. (1980) VOL.29, NO.2
C
C        AMALGAMATION OF MEANS BY THE UP-AND-DOWN BLOCKS
C        ALGORITHM OF KRUSKAL (BARLOW ET AL., 1972, P.72 )
C
      DIMENSION XO(K), WO(K), X(K), W(K), XA(K)
      DATA TOL /1.0E-6/
      IFAULT = 1
C
C        CHECK THAT K .GE. 2
C
      IF (K .LT. 2) RETURN
      IFAULT = 2
C
C        CHECK THAT THE WEIGHTS ARE POSITIVE
C
      DO 1 I = 1, K
      IF (WO(I) .LE. 0.0) RETURN
    1 CONTINUE
      IFAULT = 0
      DO 2 I = 1, K
      X(I) = XO(I)
      W(I) = WO(I)
    2 CONTINUE
      M = K
      I = 1
    3 IF (I .EQ. M) GOTO 4
      IF (X(I) .GT. X(I + 1)) GOTO 9
      IF (I .EQ. 1) GOTO 13
    4 IF (X(I - 1) .GT. X(I)) GOTO 6
      IF (I .LT. M) GOTO 13
      GOTO 14
C
C        POOL THE ACTIVE BLOCK WITH THE NEXT LOWER BLOCK
C
    6 IM1 = I - 1
      WW = W(IM1) + W(I)
      X(IM1) = (W(IM1) * X(IM1) + W(I) * X(I)) / WW
      W(IM1) = WW
      MM1 = M - 1
      IF (I .EQ. M) GOTO 8
      DO 7 J = I, MM1
      J1 = J + 1
      X(J) = X(J1)
      W(J) = W(J1)
    7 CONTINUE
    8 I = IM1
      M = MM1
      IF (M .EQ. 1) GOTO 14
      GOTO 3
C
C        POOL THE ACTIVE BLOCK WITH THE NEXT HIGHER BLOCK
C
    9 I1 = I + 1
      WW = W(I) + W(I1)
      X(I) = (W(I) * X(I) + W(I1) * X(I1)) / WW
      W(I) = WW
      MM1 = M - 1
      IF (I1 .EQ. M) GOTO 11
      DO 10 J = I1, MM1
      J1 = J + 1
      X(J) = X(J1)
      W(J) = W(J1)
   10 CONTINUE
   11 M = MM1
      IF (M .EQ. 1) GOTO 14
      IF (I .EQ. 1) GOTO 12
      IF (X(I - 1) .GT. X(I)) GOTO 6
   12 IF (I .EQ. M) GOTO 14
      IF (X(I) .GT. X(I + 1)) GOTO 9
```

```
   13 I = I + 1
      GOTO 12
C
C        OBTAIN  THE  AMALGAMATED  MEANS  XA(K) FROM  THE  WORKING
C        ARRAY  X(M)
C
   14 I1 = 1
      DO 17 I = 1, M
      S = 0.0
      DO 15 J = I1, K
      S = S + WO(J)
      XA(J) = X(I)
      IF (ABS(S - W(I)) .LT. TOL) GOTO 16
   15 CONTINUE
   16 I1 = J + 1
   17 CONTINUE
      RETURN
      END
```

Algorithm AS 150

Spectrum Estimate for a Counting Process

By DAVID M. CHARNOCK

Department of Social Sciences, W. A. Institute of Technology, Australia

Keywords: PERIODOGRAM; SPECTRUM; UNIVARIATE COUNTING PROCESS

LANGUAGE

ISO Fortran

DESCRIPTION AND PURPOSE

Given a point process observed for a period of length T in which n events occur at times $t(1)$, $t(2), \ldots, t(n)$ the subroutine first computes the normalized periodogram (see Cox and Lewis, 1966, Section 5.5)

$$I(\omega_p) = \frac{2}{n}\left|\sum_{r=1}^{n} \exp\{-i\omega_p t(r)\}\right|^2,$$

where $\omega_p = 2\pi p/T$ for $p = 1, 2, \ldots, NF$. To be able to identify important features in the spectrum, the number of frequencies at which the periodogram is computed (NF) should be greater than n. Since

$$nI(\omega_p)/2 = \left(\sum_{r=1}^{n} \cos \omega_p t(r)\right)^2 + \left(\sum_{r=1}^{n} \sin \omega_p t(r)\right)^2$$

a direct computation is very time-consuming because of the sine and cosine terms. However, for fixed $t(r)$ the sine and cosine terms at successive frequencies can be calculated recursively using the double-angle sine and cosine formulae. This considerably reduces computation time and is the procedure implemented in the subroutine. To reduce the size of accumulated errors, the sines and cosines are computed directly at every $(NRECUR)$th frequency.

Finally, an NW-point centred simple moving average of the periodogram is taken to obtain a smoothed estimate of the spectrum of the counting process. Note that the smoothed estimates and the corresponding frequencies are output in the first $(NF-NW+1)$ elements of $SPEC$ and $FREQ$.